Annotated NONMEM control file for the Eleveld PK model for Propofol
This document is an annotated NONMEM control file for the Eleveld PK model for propofol. If the annotations are removed (all lines except those in fixed-width font) then this can be run as a NONMEM control file.

The header section defines some records necessary for a NONMEM control file. The default values for $SIZES has to be changed to accommodate the large dataset for analysis. The subroutines describe the behaviour of a three compartment model in terms of volumes V1, V2, V3, clearance CL and inter-compartmental clearances Q2 and Q3.
$SIZES LTH=50 LIM6=1400

;---------------------------------------------------------------------

$PROB Eleveld propofol PK model

;---------------------------------------------------------------------

$INPUT ID DID TIME DV AMT RATE EVID AGE WGT HGT M1F2 A1V2 P1V2 STDY BMI DROP GRP

$DATA data.csv 

$SUBROUTINES ADVAN11 TRANS4
To simplify the equations later we construct indicator variables that are 1 or 0 depending on patient gender and status. We also estimate the post-menstrual-age (PMA) with the assumption that it is 40 weeks longer than the usual (post-natal) age.
$PK (ONCE PER IR)
    PAT=2-P1V2 ; indicator for patients
    VOL=P1V2-1 ; indicator for volunteers
    FEM=M1F2-1 ; indicator for females
    MAL=2-M1F2 ; indicator for males
    PMA=AGE+40/52 ; postmenstrual age in years
The development functions (ADLT) allow different characteristics between children and adults, determined by their weight (WGT). We also calculate the development function for the reference individual (AREF) in order to calculate a variable (KDEV) which is the development function normalized to the reference individual. KDEV has the value of 1 for the reference individual, a 70-kg, 35-year old, male patient. 

    SE50=THETA(13)
    SGAM=THETA(14)
    ADLT=(WGT**SGAM)/((WGT**SGAM)+(SE50**SGAM))
    CHLD=1-ADLT
    AREF=(70.**SGAM)/((70.**SGAM)+(SE50**SGAM))
    KDEV=ADLT/AREF
Only the allometric scaling exponents for V1 and V3 differ from their theoretical values of 1 for volumes and 0.75 for clearances. V1 is constant with respect to weight for adults, which is equivalent to an exponent of 0, in other words, the constant 1. V3 scales at a smaller exponent than expected and it’s value was estimated.
    V3E = THETA(17)
The aging functions are primarily focused on decreasing volumes and clearances found in the elderly and we used exponentially decreasing functions to describe these characteristics. Q2 appears to decrease comparatively rapidly with age for children but transitions to a slower rate for adults. V1 and CL appear to decrease with age at a very similar rate and improved performance under cross-validation could be achieved by sharing a single rate between V1 and CL. In Q3 and Q2 in adults the same phenomenon occurred. V3 was found to not significantly vary with age.

    KV1=EXP(-0.001*THETA(18)*(AGE-35.))
    KV2=EXP(-0.001*THETA(19)*(AGE-35.))
    KCL=KV1
    KQ2=EXP(-0.001*(AGE-35.)*(CHLD*THETA(20) + ADLT*THETA(21)))
    KQ3=EXP(-0.001*(AGE-35.)*THETA(21))
The relationship of V1 with gender was found to be a simple scaling function (KGV1) where V1 is smaller in females.

Correction of V1 for gender =

    KGV1=MAL*1 + FEM*THETA(22)
The relationship of gender with CL is more complex than with V1. CL appears to be decreased in males in the age range from about 22-69 years. Before about 22 years, there are no differences and after 69 years the relationship inverts. We used sigmoidal functions with a gamma of 8 to delineate these age ranges and the transition points were estimated from the data. We also calculate the gender correction function for CL for the reference individual (KREF) to calculate a variable (FGEN) which is the clearance adjustment function normalized to the reference individual. FGEN has the value of 1 for the reference individual, a 70-kg, 35-year old, male patient.

    CLAG=1 - (PMA**8)/((PMA**8)+(THETA(16)**8))
    PMAL=MAL*(PMA**8)/((PMA**8)+(THETA(15)**8))
    KGEN=EXP(0.01*THETA(23)*(CLAG*(1-PMAL) + (1-CLAG)*PMAL)*ADLT)
    PMR=35.+40./52.
    CREF=1 - (PMR**8)/((PMR**8)+(THETA(16)**8))
    PREF=1 * (PMR**8)/((PMR**8)+(THETA(15)**8))
    KREF=EXP(0.01*THETA(23)*(CREF*(1-PREF) + (1-CREF)*PREF)*AREF)
    FGEN=KGEN/KREF
The covariate adjustment functions (M1 though M6) are calculated to adjust the estimated parameters of the reference individual to each individual based on their weight, gender and age. They are constructed to a have a value of 1 for the reference individual, a 70-kg, 35-year old, male patient. The population typical volume and clearance parameters of the reference individual were estimated via their logarithm, not the values directly. We found that much better model fit could be achieved by estimating different typical values for patients and healthy-volunteers for all volumes and clearances.
V1 is not scaled with weight directly (a constant 1 is used) but is influenced by age, gender and development. The sigmoidal development function is influenced by weight, which allows for smaller values for children, but nearly constant values (with respect to weight) for adults. V1 shows higher population variance in patients than healthy-volunteers. NONMEM requires a scaling parameter (S1) and this was set to V1.

    M1 =1 * KV1 * KGV1 * KDEV
    V1 =EXP(VOL*(THETA(1)+ETA(1)) + PAT*(THETA(7)+ETA(2))) * M1    

    S1 =V1
V2 scales linearly with weight and is influenced by age. 

    M2 =(WGT/70.) * KV2
    V2 =EXP(VOL* THETA(2) + PAT* THETA(8) + ETA(3)) * M2

V3 scales with weight to an estimated exponent. Similarly to V1, V3 is decreased in children. It was found not to vary significantly with age.

    M3 =(WGT/70.)**V3E * KDEV
    V3 =EXP(VOL* THETA(3) + PAT* THETA(9) + ETA(4)) * M3
CL scales with weight the exponent 0.75 and is influenced by gender and age.

    M4 =(WGT/70.)**0.75 * FGEN * KCL
    CL =EXP(VOL* THETA(4) + PAT* THETA(10) + ETA(5)) * M4
We found that scaling Q2 and Q3 to the individually estimated volumes of V2 and V3 to an exponent of 0.75 lead to significantly improved model fit and predictive performance compared to the usual way of scaling them to WGT. Again, M5 and M6 are constructed to have a value of 1 for the reference individual by performing the allometric scaling on the individual estimate of V2 (or V3) normalized to their population predicted values in the reference individual (RV2 and RV3, respectively). Q2 shows a relatively rapid decline with age for small individuals, but a slower decline with age for larger individuals. For Q2 the population variance was the same in patients and healthy volunteers. 

    RV2=EXP(THETA(8))
    M5 =(V2/RV2)**0.75 * KQ2
    Q2 =EXP(VOL* THETA(5) + PAT* THETA(11) + ETA(6)) * M5
Q3 scales with V3 to an exponent 0.75 and is influenced by age. The population variance of Q3 in patients was estimated to be a very low value and was fixed to 0. 

    RV3=EXP(THETA(9))
    M6 =(V3/RV3)**0.75 * KQ3

    Q3 =EXP(VOL*(THETA(6)+ETA(7)) + PAT* THETA(12)) * M6
Proportional residual error (RESV) was calculated separately for each component study. This allows for variability in analysis methods across studies.
    RESV=1e30
    IF(STDY.EQ.1) RESV=THETA(24)/100
    IF(STDY.EQ.3) RESV=THETA(25)/100
    IF(STDY.EQ.5) RESV=THETA(26)/100
    IF(STDY.EQ.6) RESV=THETA(27)/100
    IF(STDY.EQ.7) RESV=THETA(28)/100
    IF(STDY.EQ.8) RESV=THETA(29)/100
    IF(STDY.EQ.9) RESV=THETA(30)/100
    IF(STDY.EQ.10) RESV=THETA(31)/100
    IF(STDY.EQ.11) RESV=THETA(32)/100
    IF(STDY.EQ.12) RESV=THETA(33)/100
    IF(STDY.EQ.13) RESV=THETA(34)/100
    IF(STDY.EQ.14) RESV=THETA(35)/100
    IF(STDY.EQ.15) RESV=THETA(36)/100
    IF(STDY.EQ.16) RESV=THETA(37)/100
    IF(STDY.EQ.17) RESV=THETA(38)/100
    IF(STDY.EQ.18) RESV=THETA(39)/100
    IF(STDY.EQ.19) RESV=THETA(40)/100
    IF(STDY.EQ.20) RESV=THETA(41)/100
    IF(STDY.EQ.21) RESV=THETA(42)/100
    IF(STDY.EQ.22) RESV=THETA(43)/100
    IF(STDY.EQ.23) RESV=THETA(44)/100
The following NONMEM records are the parameters and variances estimated in the NONMEM problem.
$THETA
    1.74781E+00 ; THETA 1-6 logarithm of V1,V2,V3,CL,Q2,Q3 volunteers
    2.47072E+00
    5.40370E+00
    6.03899E-01
    1.13003E+00
    7.91004E-02
    2.27971E+00 ; THETA 7-12 logarithm of V1,V2,V3,CL,Q2,Q3 patients
    3.36832E+00
    4.90135E+00
    4.26610E-01
    3.54021E-01
   -4.97825E-01
    1.66358E+01 ; THETA 13-14 parameters for development
    2.75498E+00
    2.22174E+01 ; THETA 15-16 age-parameters for gender correction
    6.91565E+01
    3.48803E-01 ; THETA 17 allometric scaling exponent for V3
    3.54508E+00 ; THETA 18 aging on V1 and CL
    8.91130E+00 ; THETA 19 aging on V2
    2.76499E+01 ; THETA 20 aging on Q2 for children
    6.34459E+00 ; THETA 21 aging on Q3 and Q2 for adults
    8.29709E-01 ; THETA 22 gender on V1
    2.25228E+01 ; THETA 23 gender on CL
    ; THETA 24-44 residual error for component studies

    4.72301E+01  4.21163E+01  4.37052E+01  5.14160E+01  3.34815E+01
    3.93393E+01  4.57665E+01  5.15078E+01  4.12550E+01  4.21820E+01  
    4.41469E+01  5.29949E+01  4.36900E+01  4.44314E+01  4.13487E+01  
    3.59628E+01  3.66601E+01  5.54676E+01  4.81950E+01  4.27194E+01
    3.69915E+01

$OMEGA
    3.169280e-02 ; volunteers V1 variance
    3.227130e-01 ; patients V1 variance
    2.937900e-01 ; V2 variance
    3.177430e-01 ; V3 variance
    6.845370e-02 ; CL variance
    1.296490e-01 ; Q2 variance
    9.143650e-02 ; volunteers Q3 variance
    4.368270e-02 ; composite error

$SIGMA
    (1 FIXED)    ; fixed error scaled for each study and individual
    1.484780e-04 ; additive error
We estimated proportional residual error (RESV) separately for each component study by fixing 1 (ERR(1)) to a variance of 1. This allows for variability in analysis methods across studies. We also allowed individuals to show differing in residual error within studies using the EXP(ETA(8)) term. An additive residual error term 2 (ERR(1)) was calculated which was shared between all individuals and component studies.

$ERROR (ONLY OBSERVATIONS)
    CP1=F 
    IPRED=CP1
    Y=CP1*(1+RESV*RESV*ERR(1)*EXP(ETA(8)))+ERR(2)
The following record instructs NONMEM to evaluate the model at the initial THETA parameters using the FOCE method with interaction. When model evaluation is done at these values the objective function value was -6263.898 but the results of a longer estimation can be slightly lower, we found -6264.542. This difference is likely due to small rounding issues.
$ESTM SIG=3 MAX=0 METHOD=1 INTERACT SORT CTYPE=4 NOABORT PRINT=1
_1446885093.unknown

_1446885147.unknown

_1446885371.unknown

_1446885385.unknown

_1446885217.unknown

_1446885103.unknown

_1446876789.unknown

_1446884950.unknown

_1446547702.unknown

_1446548167.unknown

_1446543776.unknown

